

University of New Mexico
Hypothesis Testing-2 (Fall 2018)
BIOM 505: Biostatistical Methods I (by Fares Qeadan)

Hypothesis Testing about two Population Proportions (two sample proportion z-test):
[using sample statistics]

Assumptions of this test:

- We are sampling less than 10% of the total population in each group.
 - The sample size is sufficiently large in each group such that $n_1\hat{p}_1 \geq 10$, $n_1(1 - \hat{p}_1) \geq 10$ and $n_2\hat{p}_2 \geq 10$ and $n_2(1 - \hat{p}_2) \geq 10$.
 - The two samples should be random.
 - The two samples are independent.
- (1) **Medical researchers monitoring two groups of physicians over a 6-year period found that, of 3429 doctors who took aspirin daily, 148 died from heart attack or stroke during this period. For 1710 doctors who received placebo instead of aspirin, 79 deaths were recorded. At the 0.01 level of significance, does this study indicate that taking aspirin is effective in reducing the likelihood of heart attack? Let p_1 be the true population proportion of doctors who died while taking aspirin and p_2 be the true population proportion of doctors who died while taking placebo.**

(a) The significance level α is:

(b) Give the claim as a mathematical statement:

(c) The null and alternative hypotheses are:

(d) The decision rule (about H_0) is:

(e) Conduct the test using SAS:

```
%MACRO propz2samp(n1,x1,n2,x2);
data _null_;
  file print;
  n1=&n1; x1=&x1; n2=&n2; x2=&x2;
  hatp = (x1+x2)/(n1+n2);
  hatp1 = x1/n1; hatp2 = x2/n2;
  Z2s = (hatp1 - hatp2) / sqrt(hatp*(1-hatp)*(1/n1 + 1/n2));
  ptwosided = 2*(1 - probnorm(abs(Z2s)));
  prightsided = 1 - probnorm(Z2s);
  pleftsided = probnorm(Z2s);
  L=hatp1 - hatp2-probit(0.975)*sqrt(hatp*(1-hatp)*(1/n1 + 1/n2));
  U=hatp1 - hatp2+probit(0.975)*sqrt(hatp*(1-hatp)*(1/n1 + 1/n2));
  put '==== Test for two sample proportions =====';
  put 'n1 = ' n1 ' x1 =' x1 ' pihat=' hatp1;
  put 'n2 = ' n2 ' x2 =' x2 ' p2hat=' hatp2;
  put 'Pr > |Z|: ' ptwosided pvalue.;
  put 'Pr > Z: ' prightsided pvalue.;
  put 'Pr < Z: ' pleftsided pvalue.;
  put "95% C.I.: " "(" L " - " U ")";
run;
%MEND propz2samp;
%propz2samp(3429,148,1710,79);
```

```
==== Test for two sample proportions =====
n1 = 3429  x1 =148  pihat=0.0431612715
n2 = 1710  x2 =79  p2hat=0.0461988304
Pr > |Z|: 0.6175
Pr > Z: 0.6912
Pr < Z: 0.3088
95% C.I.: (-0.014960107 - 0.0088849894 )
```

(f) Decision:

(g) Conclusion:

(h) State the error you might have made in the decision above and identify it as Type I or Type II?

Hypothesis Testing about two Population Proportions [using sample data]

- (2) Consider the Diabetes and obesity, cardiovascular risk factors data set we have used in Lab 1 (link: <http://www.mathalpha.com/lab1/diabetesfall17.sas7bdat>) to test whether the rate of diabetes among African American females is different than that of males in Virginia?

(a) The significance level α is:

(b) Give the claim as a mathematical statement:

(c) The null and alternative hypotheses are:

(d) The decision rule (about H_0) is:

(e) Conduct the test using SAS:

```
proc freq data=biom505.diabetesfall17;
table gender*diab/riskdiff(equal var=null cl=wald);
run;
```

Proportion (Risk) Difference Test	
H0: P1 - P2 = 0 Wald Method	
Proportion Difference	0.0114
ASE (H0)	0.0371
Z	0.3067
One-sided Pr > Z	0.3795
Two-sided Pr > Z	0.7591
Column 1 (diab = .)	

Column 2 Risk Estimates						
	Risk	ASE	(Asymptotic) 95% Confidence Limits		(Exact) 95% Confidence Limits	
Row 1	0.1491	0.0236	0.1029	0.1954	0.1055	0.2021
Row 2	0.1605	0.0288	0.1040	0.2170	0.1076	0.2263
Total	0.1538	0.0183	0.1180	0.1897	0.1195	0.1935
Difference	-0.0114	0.0373	-0.0844	0.0617		
Difference is (Row 1 - Row 2)						

(f) Decision:

(g) Conclusion:

(h) State the error you might have made in the decision above and identify it as Type I or Type II?

Hypothesis testing about the means of two independent populations
(assume unknown but equal population variances) [using samples statistics]
Two Independent Samples T-test

Assumptions of this test:

- The populations from which the samples have been drawn should be normal.
 - The variance of the populations should be unknown but equal i.e. $\sigma_1^2 = \sigma_2^2 = \sigma^2$, where σ^2 is unknown. This assumption can be tested formally by the F-test.
 - Samples have to be randomly drawn independent of each other. There is however no requirement that the two samples should be of equal size.
- (3) **An experiment is conducted to determine whether intensive tutoring (covering a great deal of material in a fixed amount of time) is more effective than paced tutoring (covering less material in the same amount of time) among students with special needs. Two randomly chosen groups of students with special needs are tutored separately and then administered proficiency tests. Based on the following results of the two random samples, use a significance level of 1% to verify whether intensive tutoring is more effective than paced tutoring among students with special needs? Assume that the two groups have unknown but equal variances and the proficiency scores are normal in both groups.**

	n	\bar{x}	s_x
Intensive	12	46.31	6.44
Paced	10	36.79	4.52

- (a) The significance level α is:
- (b) Give the claim as a mathematical statement:
- (c) The null and alternative hypothesis are:

(d) The decision rule (about H_0) is:

(e) Conduct the test using SAS:

```

%MACRO ttest2samp(n1,xbar1,s1,n2,xbar2,s2);
data _null_;
file print;
n1 = &n1;  xbar1 = &xbar1; s1 = &s1;  n2 = &n2;  xbar2 = &xbar2; s2 = &s2;
v1=s1**2; v2=s2**2;
f=max(of v1, v2)/min(of v1, v2);
df1=n1-1;
df2=n2-1;
dfmax=max(of df1, df2);
dfmin=min(of df1, df2);
f_p=2*(1-probf(f, dfmax, dfmin));
v_pool=((n1-1)*v1+(n2-1)*v2)/(n1+n2-2);
t_uneq=(xbar1-xbar2)/sqrt(v1/n1+v2/n2);
t_eq=(xbar1-xbar2)/sqrt(v_pool*(1/n1+1/n2));
df_uneq=(v1/n1+v2/n2)**2/((v1/n1)**2/(n1-1)+(v2/n2)**2/(n2-1));
df_eq=n1+n2-2;
t_p_uneq=2*(1-probt(abs(t_uneq), df_uneq));
prightsided_uneq = 1-probt(t_uneq, df_uneq);
pleftsided_uneq = probt(t_uneq, df_uneq);
t_p_eq=2*(1-probt(abs(t_eq), df_eq));
prightsided_eq = 1-probt(t_eq, df_eq);
pleftsided_eq = probt(t_eq, df_eq);
put  "Group      n      Mean      Std. Dev.";
put  "-----";
put  '1          ' n1  '      ' xbar1  '      ' s1;
put  '2          ' n2  '      ' xbar2  '      ' s2;
put; put;
put  "Variance    T              DF              Prob> |T|              Prob> T              Prob< T " ;
put  "-----";
put  'Unequal    ' t_uneq '      ' df_uneq '      ' t_p_uneq '      ' prightsided_uneq '      ' pleftsided_uneq;
put  'Equal      ' t_eq  '      ' df_eq  '      ' t_p_eq  '      ' prightsided_eq  '      ' pleftsided_eq;
put;
put  "For H0: Variances are equal, F=" f+5 'DF=(' dfmax+(-1)',' dfmin+(-1))' +5 'Prob>F''=' f_p;
run;
%MEND ttest2samp;

%ttest2samp(12,46.31,6.44,10,36.79,4.52);

```

Group	n	Mean	Std. Dev.
1	12	46.31	6.44
2	10	36.79	4.52

Variance	T	DF	Prob> T	Prob> T	Prob< T
Unequal	4.0596467395	19.514350657	0.00063833	0.000319165	0.999680835
Equal	3.9301851878	20	0.0008282412	0.0004141206	0.9995858794

For H_0 : Variances are equal, $F = 2.029994518$ $DF = (11,9)$ $Prob>F=0.2973407205$

(f) Decision:

(g) Conclusion:

(h) State the error you might have made in the decision above and identify it as Type I or Type II?

Hypothesis testing about the means of two independent populations,
(assume unknown and unequal population variances) [using samples statistics]
Two Independent Samples T-test

- (4) An experiment is conducted to determine whether intensive tutoring (covering a great deal of material in a fixed amount of time) is more effective than paced tutoring (covering less material in the same amount of time) among students with special needs. Two randomly chosen groups of students with special needs are tutored separately and then administered proficiency tests. Based on the following results of the two random samples, use a significance level of 1% to verify whether intensive tutoring is more effective than paced tutoring among students with special needs? Assume that the two groups have unknown and unequal variances and the proficiency scores are normal in both groups.

	n	\bar{x}	s_x
Intensive	12	46.31	6.44
Paced	10	36.79	4.52

- (a) The significance level α is:
- (b) Give the claim as a mathematical statement:
- (c) The null and alternative hypothesis are:
- (d) The decision rule (about H_0) is:

(e) Conduct the test using SAS:

```
%ttest2samp(12,46.31,6.44,10,36.79,4.52);
```

Group	n	Mean	Std. Dev.
1	12	46.31	6.44
2	10	36.79	4.52

Variance	T	DF	Prob> T	Prob> T	Prob< T
Unequal	4.0596467395	19.514350657	0.00063833	0.000319165	0.999680835
Equal	3.9301851878	20	0.0008282412	0.0004141206	0.9995858794

For H0: Variances are equal, $F = 2.029994518$ $DF = (11,9)$ $Prob>F=0.2973407205$

(f) Decision:

(g) Conclusion:

(h) State the error you might have made in the decision above and identify it as Type I or Type II?

Hypothesis testing about the means of two independent populations:

[using samples data] *Two Independent Samples T-test*

- (5) Consider the Diabetes and obesity, cardiovascular risk factors data set we have used in Lab 1 (link: <http://www.mathalpha.com/lab1/diabetesfall17.sas7bdat>) to test whether the true cholesterol mean is higher among diabetic African Americans in Virginia when compared to the non-diabetic ones?

(a) The significance level α is:

(b) Give the claim as a mathematical statement:

(c) The null and alternative hypotheses are:

(d) The decision rule (about H_0) is:

(e) Conduct the test using SAS:

```
proc ttest data=biom505.diabetesfall17 alpha=0.05;
class diab;
var chol;
run;
```

The TTEST Procedure

Variable: chol (chol)

diab	N	Mean	Std Dev	Std Err	Minimum	Maximum
0	329	203.4	41.1350	2.2678	78.0000	347.0
1	60	228.6	56.5251	7.2974	115.0	443.0
Diff (1-2)		-25.2140	43.8318	6.1531		

diab	Method	Mean	95% CL Mean	Std Dev	95% CL Std Dev
0		203.4	198.9 207.8	41.1350	38.2137 44.5437
1		228.6	214.0 243.2	56.5251	47.9126 68.9416
Diff (1-2)	Pooled	-25.2140	-37.3116 -13.1164	43.8318	40.9495 47.1540
Diff (1-2)	Satterthwaite	-25.2140	-40.4516 -9.9763		

Method	Variances	DF	t Value	Pr > t
Pooled	Equal	387	-4.10	<.0001
Satterthwaite	Unequal	70.828	-3.30	0.0015

Equality of Variances				
Method	Num DF	Den DF	F Value	Pr > F
Folded F	59	328	1.89	0.0006

(f) Decision:

(g) Conclusion:

(h) State the error you might have made in the decision above and identify it as Type I or Type II?

Hypothesis testing about the means of two dependent populations
 [using sample data] *Paired T-test*

Assumptions of this test:

- The population of differences is normally distributed.
 - The pairs are independent.
 - The sample of pairs is a random sample from its population.
- (6) **Consider the following data. These data give the systolic and diastolic blood pressure (mm Hg) for 15 patients with moderate essential hypertension, immediately before and two hours after taking a drug, captopril. The interest is in investigating the response to the drug treatment. The data is taken from Cox and Snell (1981) [applied statistics, London: Chapman and Hall]**

```
data BP;
input sbefore safter sdif dbefore dafter ddif;
CARDS;
210 201 -9 130 125 -5
169 165 -4 122 121 -1
187 166 -21 124 121 -3
160 157 -3 104 106 2
167 147 -20 112 101 -11
176 145 -31 101 85 -16
185 168 -17 121 98 -23
206 180 -26 124 105 -19
173 147 -26 115 103 -12
146 136 -10 102 98 -4
174 151 -23 98 90 -8
201 168 -33 119 98 -21
198 179 -19 106 110 4
148 129 -19 107 103 -4
154 131 -23 100 82 -18
RUN;
```

- (a) The significance level α is:

(b) Give the claim as a mathematical statement:

(c) The null and alternative hypotheses are:

(d) The decision rule (about H_0) is:

(e) Conduct the test using SAS:

```
PROC TTEST DATA=BP;  
PAIRED sbefore*safter;  
RUN;  
PROC TTEST DATA=BP;  
VAR SDIF;  
RUN;
```

```
PROC TTEST DATA=BP;  
PAIRED dbefore*dafter;  
RUN;  
PROC TTEST DATA=BP;  
VAR DDIF;  
RUN;
```

(f) Decision:

(g) Conclusion:

(h) State the error you might have made in the decision above and identify it as Type I or Type II?